

4. Gaussian derivatives

A difference which makes no difference is not a difference.

Mr. Spock (stardate 2822.3)

4.1 Introduction

We will encounter the Gaussian derivative function at many places throughout this book. The Gaussian derivative function has many interesting properties. We will discuss them in one dimension first. We study its shape and algebraic structure, its Fourier transform, and its close relation to other functions like the Hermite functions, the Gabor functions and the generalized functions. In two and more dimensions additional properties are involved like orientation (directional derivatives) and anisotropy.

4.2 Shape and algebraic structure

When we take derivatives to x (*spatial derivatives*) of the Gaussian function repetitively, we see a pattern emerging of a polynomial of increasing order, multiplied with the original (normalized) Gaussian function again. Here we show a table of the derivatives from order 0 (i.e. no differentiation) to 3.

```
<< FrontEndVision`FEV` ;
Unprotect[gauss]; gauss[x_, σ_] :=  $\frac{1}{\sigma \sqrt{2 \pi}} \text{Exp}\left[-\frac{x^2}{2 \sigma^2}\right]$ ;
Table[Factor[Evaluate[D[gauss[x, σ], {x, n}]]], {n, 0, 4}]
```

$$\left\{ \frac{e^{-\frac{x^2}{2\sigma^2}}}{\sqrt{2\pi}\sigma}, -\frac{e^{-\frac{x^2}{2\sigma^2}} x}{\sqrt{2\pi}\sigma^3}, \frac{e^{-\frac{x^2}{2\sigma^2}} (x-\sigma)(x+\sigma)}{\sqrt{2\pi}\sigma^5}, \right.$$

$$\left. -\frac{e^{-\frac{x^2}{2\sigma^2}} x (x^2 - 3\sigma^2)}{\sqrt{2\pi}\sigma^7}, \frac{e^{-\frac{x^2}{2\sigma^2}} (x^4 - 6x^2\sigma^2 + 3\sigma^4)}{\sqrt{2\pi}\sigma^9} \right\}$$

The function **Factor** takes polynomial factors apart.

The function **gauss[x,σ]** is part of the standard set of functions (in **FEV.m**) with this book, and is protected. To modify it, it must be **Unprotected**.

The zeroth order derivative is indeed the Gaussian function itself. The even order (including the zeroth order) derivative functions are even functions (i.e. symmetric around zero) and the odd order derivatives are odd functions (antisymmetric around zero). This is how the graphs of Gaussian derivative functions look like, from order 0 up to order 7 (note the marked increase in amplitude for higher order of differentiation):

```
GraphicsArray[
  Partition[Table[Plot[Evaluate[D[gauss[x, 1], {x, n}]], {x, -5, 5},
    PlotLabel -> StringJoin["Order=", ToString[n]], DisplayFunction ->
    Identity], {n, 0, 7}], 4], ImageSize -> 500] // Show;
```




Figure 4.1 Plots of the 1D Gaussian derivative function for order 0 to 7.

The Gaussian function itself is a common element of all higher order derivatives. We extract the polynomials by dividing by the Gaussian function:

$$\text{Table}\left[\text{Evaluate}\left[\frac{\text{D}[\text{gauss}[x, \sigma], \{x, n\}]}{\text{gauss}[x, \sigma]}\right], \{n, 0, 4\}\right] // \text{Simplify}$$

$$\left\{1, -\frac{x}{\sigma^2}, \frac{x^2 - \sigma^2}{\sigma^4}, -\frac{x^3 - 3x\sigma^2}{\sigma^6}, \frac{x^4 - 6x^2\sigma^2 + 3\sigma^4}{\sigma^8}\right\}$$

These polynomials have the same order as the derivative they are related to. Note that the highest order of x is the same as the order of differentiation, and that we have a plus sign for the highest order of x for even number of differentiation, and a minus signs for the odd orders.

These polynomials are the Hermite polynomials, called after Charles Hermite, a brilliant French mathematician (see figure 4.2).

```
Show[Import["Charles Hermite.jpg"], ImageSize -> 150];
```



Figure 4.2 Charles Hermite (1822-1901).

They emerge from the following definition: $\frac{\partial^n e^{-x^2}}{\partial x^n} = (-1)^n H_n(x) e^{-x^2}$. The function $H_n(x)$ is the Hermite polynomial, where n is called the order of the polynomial. When we make the substitution $x \rightarrow x/(\sigma\sqrt{2})$, we get the following relation between the Gaussian function $G(x, \sigma)$ and its derivatives: $\frac{\partial^n G(x, \sigma)}{\partial x^n} = (-1)^n \frac{1}{(\sigma\sqrt{2})^n} H_n\left(\frac{x}{\sigma\sqrt{2}}\right) G(x, \sigma)$.

In *Mathematica* the function H_n is given by the function **HermiteH[n, x]**. Here are the Hermite functions from zeroth to fifth order:

```
Table[HermiteH[n, x], {n, 0, 7}] // TableForm
```

```
1
2 x
-2 + 4 x^2
-12 x + 8 x^3
12 - 48 x^2 + 16 x^4
120 x - 160 x^3 + 32 x^5
-120 + 720 x^2 - 480 x^4 + 64 x^6
-1680 x + 3360 x^3 - 1344 x^5 + 128 x^7
```

The inner scale σ is introduced in the equation by substituting $x \rightarrow \frac{x}{\sigma\sqrt{2}}$. As a consequence, with each differentiation we get a new factor $\frac{1}{\sigma\sqrt{2}}$. So now we are able to calculate the 1-D Gaussian derivative functions **gd[x, n, σ]** directly with the Hermite polynomials, again incorporating the normalization factor $\frac{1}{\sigma\sqrt{2}\pi}$:

```
Clear[σ];
gd[x_, n_, σ_] :=  $\left(\frac{-1}{\sigma\sqrt{2}}\right)^n$  HermiteH[n,  $\frac{x}{\sigma\sqrt{2}}$ ]  $\frac{1}{\sigma\sqrt{2}\pi}$  Exp[- $\frac{x^2}{2\sigma^2}$ ]
```

Check:

```
Simplify[gd[x, 4, σ], σ > 0]
 $\frac{e^{-\frac{x^2}{2\sigma^2}} (x^4 - 6x^2\sigma^2 + 3\sigma^4)}{\sqrt{2}\pi\sigma^9}$ 
Simplify[D[ $\frac{1}{\sigma\sqrt{2}\pi}$  Exp[- $\frac{x^2}{2\sigma^2}$ ], {x, 4}], σ > 0]
 $\frac{e^{-\frac{x^2}{2\sigma^2}} (x^4 - 6x^2\sigma^2 + 3\sigma^4)}{\sqrt{2}\pi\sigma^9}$ 
```

The amplitude of the Hermite polynomials explodes for large x , but the Gaussian envelop suppresses *any* polynomial function. No matter how high the polynomial order, the exponential function always wins. We can see this graphically when we look at e.g. the 7th order Gaussian derivative without (i.e. the Hermite function, figure left) and with its Gaussian weight function (figure right). Note the vertical scales:

```

f[x_] :=  $\left(\frac{1}{\sqrt{2}}\right)^7 \text{HermiteH}[7, \frac{x}{\sqrt{2}}];
\text{DisplayTogetherArray}[\{\text{Plot}[f[x], \{x, -5, 5\}],
\text{p2} = \text{Plot}[f[x] \text{Exp}[-\frac{x^2}{2}], \{x, -5, 5\}], \text{ImageSize} \rightarrow 400];$ 
```




Figure 4.3 Left: The 7th order Hermite polynomial. Right: idem, with a Gaussian envelop (weighting function). This is the 7th order Gaussian derivative kernel.

Due to the limiting extent of the Gaussian window function, the amplitude of the Gaussian derivative function can be negligible at the location of the larger zeros. We plot an example, showing the 20th order derivative and its Gaussian envelope function:

```

n = 20; σ = 1; \text{DisplayTogether}[\{\text{FilledPlot}[gd[x, n, σ], \{x, -5, 5\}],
\text{Plot}[\frac{gd[0, n, σ] \text{gauss}[x, σ]}{\text{gauss}[0, σ]}, \{x, -5, 5\}], \text{ImageSize} \rightarrow 200];

```




Figure 4.4 The 20th order Gaussian derivative's outer zero-crossings vanish in negligence. Note also that the amplitude of the Gaussian derivative function is not bounded by the Gaussian window. The Gabor kernels, as we will discuss later in section 4.7, are bounded by the Gaussian window.

How fast the Gaussian function goes zero can be seen from its values at $x = 3\sigma$, $x = 4\sigma$ and $x = 5\sigma$, relative to its peak value:

```

\text{Table}[\frac{\text{gauss}[\sigma, 1]}{\text{gauss}[0, 1]}, \{\sigma, 3, 5\}] // \text{N}
\{0.011109, 0.000335463, 3.72665 \times 10^{-6}\}

```

and in the limit:

```
Limit[gd[x, 7, 1], x -> Infinity]
```

```
0
```

The Hermite polynomials belong to the family of orthogonal functions on the infinite interval $(-\infty, \infty)$ with the weight function e^{-x^2} , $\int_{-\infty}^{\infty} e^{-x^2} H_n(x) H_m(x) dx = 2^{n+m} n! \sqrt{\pi} \delta_{nm}$, where δ_{nm} is the Kronecker delta, or delta tensor. $\delta_{nm} = 1$ for $n = m$, and $\delta_{nm} = 0$ for $n \neq m$.

```
Table[ $\int_{-\infty}^{\infty} \text{Exp}[-x^2] \text{HermiteH}[k, x] \text{HermiteH}[m, x] dx,$   
{k, 0, 3}, {m, 0, 3}] // MatrixForm
```

$$\begin{pmatrix} \sqrt{\pi} & 0 & 0 & 0 \\ 0 & 2\sqrt{\pi} & 0 & 0 \\ 0 & 0 & 8\sqrt{\pi} & 0 \\ 0 & 0 & 0 & 48\sqrt{\pi} \end{pmatrix}$$

The Gaussian derivative functions, with their weight function $e^{-\frac{x^2}{2}}$ are *not* orthogonal. We check this with some examples:

$$\left\{ \int_{-\infty}^{\infty} \text{gd}[x, 2, 1] \text{gd}[x, 3, 1] dx, \int_{-\infty}^{\infty} \text{gd}[x, 2, 1] \text{gd}[x, 4, 1] dx \right\}$$

$$\left\{ 0, -\frac{15}{16\sqrt{\pi}} \right\}$$

Other families of orthogonal polynomials are e.g. Legendre, Chebyshev, Laguerre, and Jacobi polynomials. Other orthogonal families of functions are e.g. Bessel functions and the spherical harmonic functions. The area under the Gaussian derivative functions is *not* unity, e.g. for the first derivative:

```
SetOptions[Integrate, GenerateConditions -> False];
```

$$\int_0^{\infty} \text{gd}[x, 1, \sigma] dx$$

$$-\frac{1}{\sqrt{2\pi}}$$

4.3 Gaussian derivatives in the Fourier domain

The Fourier transform of the derivative of a function is $(-i\omega)$ times the Fourier transform of the function. For each differentiation, a new factor $(-i\omega)$ is added. So the Fourier transforms of the Gaussian function and its first and second order derivatives are:

```
 $\sigma = .;$  Simplify[FourierTransform[  
{gauss[x,  $\sigma$ ],  $\partial_x$  gauss[x,  $\sigma$ ],  $\partial_{\{x,2\}}$  gauss[x,  $\sigma$ ]}, x,  $\omega$ ],  $\sigma > 0$ ]
```

$$\left\{ \frac{e^{-\frac{1}{2}\sigma^2\omega^2}}{\sqrt{2\pi}}, -\frac{i e^{-\frac{1}{2}\sigma^2\omega^2} \omega}{\sqrt{2\pi}}, -\frac{e^{-\frac{1}{2}\sigma^2\omega^2} \omega^2}{\sqrt{2\pi}} \right\}$$

In general: $\mathcal{F} \left\{ \frac{\partial^n G(x, \sigma)}{\partial x^n} \right\} = (-i\omega)^n \mathcal{F} \{G(x, \sigma)\}$.

Gaussian derivative kernels also act as bandpass filters. The maximum is at $\omega = \sqrt{n}$:

$$\mathbf{n = .; \sigma = 1; Solve[Evaluate[D[\frac{(-I \omega)^n \text{Exp}[-\frac{\sigma^2 \omega^2}{2}]}{\sqrt{2 \pi}}], \omega]] == 0, \omega}$$

$$\{\{\omega \rightarrow i 0^{\frac{1}{n}}\}, \{\omega \rightarrow -\sqrt{n}\}, \{\omega \rightarrow \sqrt{n}\}\}$$

The normalized powerspectra show that higher order of differentiation means a higher center frequency for the *bandpass* filter. The bandwidth remains virtually the same.

```

$$\sigma = 1; \mathbf{p1 = Table[Plot[Abs[\frac{(-I \omega)^n \text{Exp}[-\frac{\sigma^2 \omega^2}{2}]}{\sqrt{2 \pi}}], \omega], \{n, 1, 12\}],$$

```

$$\{\omega, 0, 6\}, \text{DisplayFunction} \rightarrow \text{Identity}], \{n, 1, 12\};$$

```
Show[p1, DisplayFunction -> $DisplayFunction, PlotRange -> All,
  AxesLabel -> {"\omega", ""}, ImageSize -> 400];
```




Figure 4.5 Normalized power spectra for Gaussian derivative filters for order 1 to 12, lowest order is left-most graph, $\sigma = 1$. Gaussian derivative kernels act like bandpass filters.

- ▲ Task 4.1 Show with partial integration and the definitions from section 3.10 that the Fourier transform of the derivative of a function is $(-i\omega)$ times the Fourier transform of the function.
- ▲ Task 4.2 Note that there are several definitions of the signs occurring in the Fourier transform (see the Help function in *Mathematica* under Fourier). Show that with the other definitions it is possible to arrive to the result that the Fourier

transform of the derivative of a function is $(i\omega)$ times the Fourier transform of the function. In this book we stick to the default definition.

4.4 Zero crossings of Gaussian derivative functions

```
gd[x_, n_, σ_] :=  $\left(\frac{-1}{\sigma\sqrt{2}}\right)^n \text{HermiteH}\left[n, \frac{x}{\sigma\sqrt{2}}\right] \frac{1}{\sigma\sqrt{2\pi}} \text{Exp}\left[-\frac{x^2}{2\sigma^2}\right];$ 
nmax = 20; σ = 1;
Show[Graphics[Flatten[Table[{PointSize[0.015], Point[{n, x}]} /.
    Solve[HermiteH[n,  $\frac{x}{\sqrt{2}}$ ] == 0, x], {n, 1, nmax}], 1]],
    AxesLabel -> {"Order", "Zeros of\nHermiteH"}, Axes -> True,
    ImageSize -> 350];
```




Figure 4.6 Zero crossings of Gaussian derivative functions to 20th order. Each dot is a zero-crossing.

How *wide* is a Gaussian derivative? This may seem a non-relevant question, because the Gaussian envelop often completely determines its behaviour. However, the number of zero-crossings is equal to the order of differentiation, because the Gaussian weighting function is a positive definite function.

It is of interest to study the behaviour of the zero-crossings. They move further apart with higher order. We can define the 'width' of a Gaussian derivative function as the distance between the outermost zero-crossings. The zero-crossings of the Hermite polynomials determine the zero-crossings of the Gaussian derivatives. In figure 4.6 all zeros of the first 20 Hermite functions as a function of the order are shown. Note that the zeros of the second derivative are just one standard deviation from the origin:

```
σ = .; Simplify[Solve[D[gauss[x, σ], {x, 2}] == 0, x], σ > 0]
{{x -> -σ}, {x -> σ}}
```

An exact analytic solution for the largest zero is not known. The formula of Zernicke (1931) specifies a range, and Szego (1939) gives a better estimate:

```
Block[{$DisplayFunction = Identity},
  p1 = Plot[2 Sqrt[n + 1 - 3.05  $\sqrt[3]{n+1}$ ], {n, 5, 50}];
  (* Zernicke upper limit *)
  p2 = Plot[2 Sqrt[n + 1 - 1.15  $\sqrt[3]{n+1}$ ], {n, 1, 50}];
  (* Zernicke lower limit *)
  p3 = Plot[2  $\sqrt{n+.5}$  - 2.338098 /  $\sqrt[6]{n+.5}$ ,
    {n, 1, 50}, PlotStyle -> Dashing[ {.01, .02} ]];
  Show[{p1, p2, p3}, AxesLabel -> {"Order",
    "Width of Gaussian\nderivative (in  $\sigma$ )"}, ImageSize -> 260];
```




Figure 4.7 Estimates for the width of Gaussian derivative functions to 50th order. Width is defined as the distance between the outmost zero-crossings. Top and bottom graph: estimated range by Zernicke (1931), dashed graph: estimate by Szego (1939).

For very high orders of differentiation of course the numbers of zero-crossings increases, but also their mutual distance between the zeros becomes more equal. In the limiting case of infinite order the Gaussian derivative function becomes a sinusoidal function:

$$\lim_{n \rightarrow \infty} \frac{\partial^n G}{\partial^n x}(x, \sigma) = \text{Sin}\left(x \sqrt{\frac{1}{\sigma} \left(\frac{n+1}{2}\right)}\right).$$

4.5 The correlation between Gaussian derivatives

Higher order Gaussian derivative kernels tend to become more and more similar. This makes them not very suitable as a basis. But before we investigate their role in a possible basis, let us investigate their similarity.

In fact we can express exactly how much they resemble each other as a function of the difference in differential order, by calculating the *correlation* between them. We derive the correlation below, and will appreciate the nice mathematical properties of the Gaussian function. Because the higher dimensional Gaussians are just the product of 1D Gaussian functions, it suffices to study the 1D case.

Compare e.g. the 20th and 24nd derivative function:

```
Block[{$DisplayFunction = Identity},
  g1 = Plot[gd[x, 20, 2], {x, -7, 7}, PlotLabel -> "Order 20"];
  g2 = Plot[gd[x, 24, 2], {x, -7, 7}, PlotLabel -> "Order 24"];
  Show[GraphicsArray[{g1, g2}], ImageSize -> 400];
```




Figure 4.8 Gaussian derivative functions start to look more and more alike for higher order. Here the graphs are shown for the 20th and 24th order of differentiation.

The correlation coefficient between two functions is defined as the integral of the product of the functions over the full domain (in this case $-\infty$ to $+\infty$).

Because we want the coefficient to be unity for complete correlation (when the functions are identical by an amplitude scaling factor) we divide the coefficient by the so-called autocorrelation coefficients, i.e. the correlation of the functions with themselves.

We then get as definition for the correlation coefficient r between two Gaussian derivatives of order n and m :

$$r_{n,m} = \frac{\int_{-\infty}^{\infty} g^{(n)}(x) g^{(m)}(x) dx}{\sqrt{\int_{-\infty}^{\infty} [g^{(n)}(x)]^2 dx \int_{-\infty}^{\infty} [g^{(m)}(x)]^2 dx}}$$

with $g^{(n)}(x) = \frac{\partial^n g(x)}{\partial x^n}$. The Gaussian kernel $g(x)$ itself is an even function, and, as we have seen before, $g^{(n)}(x)$ is an even function for n is even, and an odd function for n is odd. The correlation between an even function and an odd function is zero. This is the case when n and m are both not even or both not odd, i.e. when $(n - m)$ is odd. We now can see already two important results:

$$\begin{aligned} r_{n,m} &= 0 \text{ for } (n - m) \text{ odd;} \\ r_{n,m} &= 1 \text{ for } n = m. \end{aligned}$$

The remaining case is when $(n - m)$ is even. We take $n > m$. Let us first look to the nominator, $\int_{-\infty}^{\infty} g^{(n)}(x) g^{(m)}(x) dx$. The standard approach to tackle high exponents of functions in integrals, is the reduction of these exponents by partial integration:

$$\begin{aligned} \int_{-\infty}^{\infty} g^{(n)}(x) g^{(m)}(x) dx &= \\ g^{(n-1)}(x) g^{(m)}(x) \Big|_{-\infty}^{\infty} - \int_{-\infty}^{\infty} g^{(n-1)}(x) g^{(m+1)}(x) dx &= (-1)^k \int_{-\infty}^{\infty} g^{(n-k)}(x) g^{(m+k)}(x) dx \end{aligned}$$

when we do the partial integration k times. The 'stick expression' $g^{(n-1)}(x) g^{(m)}(x) \Big|_{-\infty}^{\infty}$ is zero because any Gaussian derivative function goes to zero for large x . We can choose k such that the exponents in the integral are equal (so we end up with the square of a Gaussian derivative

function). So we make $(n - k) = (m + k)$, i.e. $k = \frac{(n-m)}{2}$. Because we study the case that $(n - m)$ is even, k is an integer number. We then get:

$$(-1)^k \int_{-\infty}^{\infty} g^{(n-k)}(x) g^{(m+k)}(x) dx = (-1)^{\frac{n-m}{2}} \int_{-\infty}^{\infty} g^{(\frac{n+m}{2})}(x) g^{(\frac{n+m}{2})}(x) dx$$

The *total energy* of a function in the spatial domain is the integral of the square of the function over its full extent. The famous theorem of Parseval states that the total energy of a function in the spatial domain is equal to the total energy of the function in the Fourier domain, i.e. expressed as the integral of the square of the Fourier transform over its full extent. Therefore

$$\begin{aligned} (-1)^{\frac{n-m}{2}} \int_{-\infty}^{\infty} g^{(\frac{n+m}{2})}(x) g^{(\frac{n+m}{2})}(x) dx &\stackrel{\text{Parseval}}{=} (-1)^{\frac{n-m}{2}} \frac{1}{2\pi} \int_{-\infty}^{\infty} |(i\omega)^{\frac{n+m}{2}} \hat{g}(\omega)|^2 d\omega = \\ &(-1)^{\frac{n-m}{2}} \frac{1}{2\pi} \int_{-\infty}^{\infty} \omega^{n+m} \hat{g}^2(\omega) d\omega = (-1)^{\frac{n-m}{2}} \frac{1}{2\pi} \int_{-\infty}^{\infty} \omega^{n+m} e^{-\sigma^2 \omega^2} d\omega \end{aligned}$$

We now substitute $\omega' = \sigma \omega$, and get finally: $(-1)^{\frac{n-m}{2}} \frac{1}{2\pi} \frac{1}{\sigma^{n+m+1}} \int_{-\infty}^{\infty} \omega'^{(n+m)} e^{-\omega'^2} d\omega'$.

This integral can be looked up in a table of integrals, but why not let *Mathematica* do the job (we first clear n and m):

```
Clear[n, m]; Integrate[x^{m+n} e^{-x^2} dx, x]
1/2 (1 + (-1)^{m+n}) Gamma[1/2 (1 + m + n)] Log[e]^{1/2 (-1-m-n)}
```

The function **Gamma** is the Euler gamma function. In our case $\text{Re}[m+n] > -1$, so we get for our correlation coefficient for $(n - m)$ even:

$$r_{n,m} = \frac{(-1)^{\frac{n-m}{2}} \frac{1}{2\pi} \frac{1}{\sigma^{n+m+1}} \Gamma(\frac{m+n+1}{2})}{\sqrt{\frac{1}{2\pi} \frac{1}{\sigma^{2n+1}} \Gamma(\frac{2n+1}{2})} \frac{1}{2\pi} \frac{1}{\sigma^{2m+1}} \Gamma(\frac{2m+1}{2})} = \frac{(-1)^{\frac{n-m}{2}} \Gamma(\frac{m+n+1}{2})}{\sqrt{\Gamma(\frac{2n+1}{2}) \Gamma(\frac{2m+1}{2})}}$$

Let's first have a look at this function for a range of values for n and m (0-15):

```
r[n_, m_] := (-1)^{n-m/2} Gamma[(m+n+1)/2] / Sqrt[Gamma[(2n+1)/2] Gamma[(2m+1)/2]];
ListPlot3D[Table[Abs[r[n, m]], {n, 0, 15}, {m, 0, 15}],
  Axes -> True, AxesLabel -> {"n", "m", "Abs[r[n,m]"]},
  ViewPoint -> {-2.348, -1.540, 1.281}, ImageSize -> 220];
```




Figure 4.9 The magnitude of the correlation coefficient of Gaussian derivative functions for $0 < n < 15$ and $0 < m < 15$. The origin is in front.

Here is the function tabulated:

```
Table[NumberForm[r[n, m] // N, 3], {n, 0, 4}, {m, 0, 4}] // MatrixForm
```

$$\begin{pmatrix} 1. & 0. - 0.798 i & -0.577 & 0. + 0.412 i & 0.293 \\ 0. + 0.798 i & 1. & 0. - 0.921 i & -0.775 & 0. + 0.623 i \\ -0.577 & 0. + 0.921 i & 1. & 0. - 0.952 i & -0.845 \\ 0. - 0.412 i & -0.775 & 0. + 0.952 i & 1. & 0. - 0.965 i \\ 0.293 & 0. - 0.623 i & -0.845 & 0. + 0.965 i & 1. \end{pmatrix}$$

The correlation is unity when $n = m$, as expected, is negative when $n - m = 2$, and is positive when $n - m = 4$, and is complex otherwise. Indeed we see that when $n - m = 2$ the functions are even but of opposite sign:

```
Block[{$DisplayFunction = Identity},
  p1 = Plot[gd[x, 20, 2], {x, -5, 5}, PlotLabel -> "Order 20"];
  p2 = Plot[gd[x, 22, 2], {x, -5, 5}, PlotLabel -> "Order 22"];
  Show[GraphicsArray[{p1, p2}], ImageSize -> 450];
```




Figure 4.10 Gaussian derivative functions differing two orders are of opposite polarity.

and when $n - m = 1$ they have a phase-shift, leading to a complex correlation coefficient:

```
Block[{$DisplayFunction = Identity},
  p1 = Plot[gd[x, 20, 2], {x, -5, 5}, PlotLabel -> "Order 20"];
  p2 = Plot[gd[x, 21, 2], {x, -5, 5}, PlotLabel -> "Order 21"];
  Show[GraphicsArray[{p1, p2}], ImageSize -> 450];
```




Figure 4.11 Gaussian derivative functions differing one order display a phase shift.

Of course, this is easy understood if we realize the factor $(-i\omega)$ in the Fourier domain, and that $i = e^{-i\frac{\pi}{2}}$. We plot the behaviour of the correlation coefficient of two close orders for large n . The asymptotic behaviour towards unity for increasing order is clear.

```
Plot[-r[n, n + 2], {n, 1, 20}, DisplayFunction -> $DisplayFunction,
  AspectRatio -> .4, PlotRange -> {.8, 1.01},
  AxesLabel -> {"Order", "Correlation\ncoefficient"}];
```




Figure 4.12 The correlation coefficient between a Gaussian derivative function and its ever neighbour up quite quickly tends to unity for high differential order.

4.6 Discrete Gaussian kernels

```
 $\sigma = 2$ ; Plot[ {  $\frac{1}{\sqrt{2\pi\sigma^2}} \text{Exp}[\frac{-x^2}{2\sigma^2}]$ ,  $\frac{1}{\sqrt{2\pi\sigma^2}} \text{BesselI}[x, \sigma^2] / \text{BesselI}[0, \sigma^2]$  },
  {x, 0, 8}, PlotStyle -> {RGBColor[0, 0, 0], Dashing[{0.02, 0.02}]},
  PlotLegend -> {"Gauss", "Bessel"}, LegendPosition -> {1, 0},
  LegendLabel -> " $\sigma = 2$ ", PlotRange -> All, ImageSize -> 400];
```




Figure 4.13 The graphs of the Gaussian kernel and the modified Bessel function of the first kind are very alike.

Lindeberg [Lindeberg1990] derived the optimal kernel for the case when the Gaussian kernel was discretized and came up with the "modified Bessel function of the first kind". In *Mathematica* this function is available as **BesselI**. This function is almost equal to the Gaussian kernel for $\sigma > 1$, as we see in the figure on the previous page. Note that the Bessel function has to be normalized by its value at $x = 0$. For larger σ the kernels become rapidly very similar.

4.7 Other families of kernels

The first principles that we applied to derive the Gaussian kernel in chapter 2 essentially stated "we know nothing" (at this stage of the observation). Of course, we can relax these principles, and introduce some knowledge. When we want to derive a set of apertures tuned to a *specific spatial frequency* \vec{k} in the image, we add this physical quantity to the matrix of the dimensionality analysis:

```
m = {{1, -1, -2, -2, -1}, {0, 0, 1, 1, 0}};
TableForm[m,
  TableHeadings -> {{"meter", "candela"}, {""σ", "ω", "L0", "L", "k"}}]
```

	σ	ω	L0	L	k
meter	1	-1	-2	-2	-1
candela	0	0	1	1	0

The nullspace is now:

```
NullSpace[m]
{{1, 0, 0, 0, 1}, {0, 0, -1, 1, 0}, {1, 1, 0, 0, 0}}
```

Following the exactly similar line of reasoning, we end up from this new set of constraints with a new family of kernels, the *Gabor family of receptive fields*, with are given by a sinusoidal function (at the specified spatial frequency) under a Gaussian window.

In the Fourier domain: $Gabor(\omega, \sigma, k) = e^{-\omega^2 \sigma^2} e^{ik\omega}$, which translates into the spatial domain:

$$\mathbf{gabor}[\mathbf{x}_-, \sigma_-] := \mathbf{Sin}[\mathbf{x}] \frac{1}{\sqrt{2 \pi \sigma^2}} \mathbf{Exp}\left[-\frac{\mathbf{x}^2}{2 \sigma^2}\right];$$

The Gabor function model of cortical receptive fields was first proposed by Marcelja in 1980 [Marcelja1980]. However the functions themselves are often credited to Gabor [Gabor1946] who supported their use in communications.

Gabor functions are defined as the sinus function under a Gaussian window with scale σ . The phase ϕ of the sinusoidal function determines its detailed behaviour, e.g. for $\phi = \pi/2$ we get an even function. Gabor functions can look very much like Gaussian derivatives, but there are essential differences:

- Gabor functions have an infinite number of zero-crossings on their domain.
- The amplitudes of the sinusoidal function never exceeds the Gaussian envelope.

```

gabor[x_, φ_, σ_] := Sin[x + φ] gauss[x, σ];
Block[{$DisplayFunction = Identity, p, pg},
  p = Plot[gabor[x, #, 10], {x, -30, 30}, PlotRange -> {-.04, .04}] &;
  pg = Plot[gauss[x, 10], {x, -30, 30}, PlotRange -> {-.04, .04}];
  p13 = Show[p[0], pg]; p23 = Show[p[π/2], pg];

Show[GraphicsArray[{p13, p23}], ImageSize -> 450];

```




Figure 4.14 Gabor functions are sinusoidal functions with a Gaussian envelope. Left: $\text{Sin}[x] G[x,10]$; right: $\text{Sin}[x+\pi/2] G[x,10]$.

Gabor functions can be made to look very similar by an appropriate choice of parameters:

```

σ = 1; gd[x_, σ_] = D[ $\frac{1}{\sqrt{2 \pi \sigma^2}} \text{Exp}[-\frac{x^2}{2 \sigma^2}]$ , x];
Plot[{-1.2 gabor[x, 1.2], gd[x, 1]}, {x, -4, 4},
  PlotStyle -> {Dashing[{0.02, 0.02]}, RGBColor[0, 0, 0]},
  PlotLegend -> {"Gabor", "Gauss"},
  LegendPosition -> {1.2, -0.3}, ImageSize -> 320];

```




Figure 4.15 Gabor functions can be made very similar to Gaussian derivative kernels. In a practical application then there is no difference in result. Dotted graph: Gaussian first derivative kernel. Continuous graph: Minus the Gabor kernel with the same σ as the Gaussian kernel. Note the necessity of sign change due to the polarity of the sinusoidal function.

If we relax one or more of the first principles (leave one or more out, or add other axioms), we get other families of kernels. E.g. when we add the constraint that the kernel should be tuned to a specific spatial frequency, we get the family of *Gabor kernels* [Florack1992a, Florack1997a]. It was recently shown by Duits et al. [Duits2002a], extending the work of Pauwels [Pauwels1995], that giving up the constraint of separability gives a new family of interesting *Poisson* scale-space kernels, defined by the solution of the Dirichlet problem

$\frac{\partial L}{\partial s} = -(-\Delta)^\alpha L$. For $\alpha = 1$ we find the Gaussian scale-space, for $\alpha = \frac{1}{2}$ we get the Poisson scale-space. In this book we limit ourselves to the Gaussian kernel.

We conclude this section by the realization that the front-end visual system at the retinal level must be uncommitted, no feedback from higher levels is at stake, so the Gaussian kernel seems a good candidate to start observing with at this level. At higher levels this constraint is released.

The extensive feedback loops from the primary visual cortex to the LGN may give rise to 'geometry-driven diffusion' [TerHaarRomeny1994f], nonlinear scale-space theory, where the early differential geometric measurements through e.g. the simple cells may modify the kernels LGN levels. Nonlinear scale-space theory will be treated in chapter 21.

- ▲ Task 4.2 When we have noise in the signal to be differentiated, we have two counterbalancing effect when we change differential order and scale: for higher order the noise is amplified (the factor $(-i\omega)^n$ in the Fourier transform representation) and the noise is averaged out for larger scales. Give an explicit formula in our *Mathematica* framework for the propagation of noise when filtered with Gaussian derivatives. Start with the easiest case, i.e. pixel-uncorrelated (white) noise, and continue with correlated noise. See for a treatment of this subject the work by Blom et al. [Blom1993a].
- ▲ Task 4.3 Give an explicit formula in our *Mathematica* framework for the propagation of noise when filtered with a *compound function* of Gaussian derivatives, e.g. by the Laplacian $\frac{\partial^2 G}{\partial x^2} + \frac{\partial^2 G}{\partial y^2}$. See for a treatment of this subject the work by Blom et al. [Blom1993a].

4.8 Higher dimensions and separability

Gaussian derivative kernels of higher dimensions are simply made by multiplication. Here again we see the separability of the Gaussian, i.e. this *is* the separability. The function **gd2D[x, y, n, m, σx, σy]** is an example of a Gaussian partial derivative function in 2D, first order derivative to x , second order derivative to y , at scale 2 (equal for x and y):

```
gd2D[x_, y_, n_, m_, σx_, σy_] := gd[x, n, σx] gd[y, m, σy];
Plot3D[gd2D[x, y, 1, 2, 2, 2], {x, -7, 7},
  {y, -7, 7}, AxesLabel -> {x, y, ""}, PlotPoints -> 40,
  PlotRange -> All, Boxed -> False, Axes -> True, ImageSize -> 190];
```




Figure 4.16 Plot of $\frac{\partial^3 G(x,y)}{\partial x \partial y^2}$. The two-dimensional Gaussian derivative function can be constructed as the product of two one-dimensional Gaussian derivative functions, and so for higher dimensions, due to the separability of the Gaussian kernel for higher dimensions.

The ratio $\frac{\sigma_x}{\sigma_y}$ is called the anisotropy ratio. When it is unity, we have an *isotropic* kernel, which diffuses in the x and y direction by the same amount. The Greek word 'isos' (ἴσος) means 'equal', the Greek word 'tropos' (τροπος) means 'direction' (the Greek word 'topos' (τοπος) means 'location, place').

In 3D the iso-intensity surfaces of the Gaussian kernels are shown (and can be interactively manipulated) with the command `MVContourPlot3D` from the OpenGL viewer 'MathGL3D' by J.P.Kuska

(phong.informatik.uni-leipzig.de/~kuska/mathgl3dv3):

```
<< MathGL3d`OpenGLViewer` ;

MVClear[]; σ = 1;

p1 = Table[MVContourPlot3D[Evaluate[D[E- $\frac{x^2+y^2+z^2}{2\sigma^2}$ , {x, n}]], {x, -6, 6},
  {y, -4, 4}, {z, -3, 0}, Contours -> Range[-.6, .6, .1], PlotPoints -> 60,
  BoxRatios -> {2, 2, 1}, DisplayFunction -> Identity], {n, 1, 3}];
Show[GraphicsArray[p1], ImageSize -> 400];
```




Figure 4.17 Iso-intensity surface for Gaussian derivative kernels in 3D. Left: $\frac{\partial G}{\partial x}$; middle: $\frac{\partial^2 G}{\partial x^2}$; right: $\frac{\partial^3 G}{\partial x^3}$.

The sum of 2 of the three 2nd order derivatives is called the 'hotdog' detector:

```
MVclear[]; σ = 1;
p1 = MVContourPlot3D[Evaluate[∂x,x E- $\frac{x^2+y^2+z^2}{2\sigma^2}$  + ∂z,z E- $\frac{x^2+y^2+z^2}{2\sigma^2}$ ], {x, -6, 6},
{y, -4, 4}, {z, -3, 0}, Contours → Range[-.6, .6, .1],
PlotPoints → 60, BoxRatios → {2, 2, 1}, ImageSize -> 150];
```




Figure 4.18 Iso-intensity surface for the Gaussian derivative in 3D $\frac{\partial^2 G}{\partial x^2} + \frac{\partial^2 G}{\partial z^2}$.

4.9 Summary of this chapter

The Gaussian derivatives are characterized by the product of a polynomial function, the Hermite polynomial, and a Gaussian kernel. The order of the Hermite polynomial is the same as the differential order of the Gaussian derivative. Many interesting recursive relations exist for Hermite polynomials, making them very suitable for analytical treatment. The shape of the Gaussian derivative kernel can be very similar to specific Gabor kernels. One essential difference is the number of zerocrossing: this is always infinite for Gabor kernels, the number of zerocrossings of Gaussian derivatives is equal to the differential order. The envelope of the Gaussian derivative amplitude is not the Gaussian function, as is the case for Gabor kernels.

The even orders are symmetric kernels, the odd orders are asymmetric kernels. The normalized zeroth order kernel has unit area by definition, the Gaussian derivative kernels of the normalized kernel have no unit area.

Gaussian derivatives are not orthogonal kernels. They become more and more correlated for higher order, if odd or even specimens are compared. The limiting case for infinite order leads to a sinusoidal (for the odd orders) or cosinusoidal (for the even orders) function with a Gaussian envelope, i.e. a Gabor function.

In the vision chapters we will encounter the Gaussian derivative functions as suitable and likely candidates for the receptive fields in the primary visual cortex.